職員配置スケジュール表VBA（2025年更新版）

Option Explicit

' 列位置の定数
Private Const COL_アルファベット As Long = 2
Private Const COL_勤務開始時 As Long = 3
Private Const COL_勤務開始分 As Long = 5
Private Const COL_勤務終了時 As Long = 7
Private Const COL_勤務終了分 As Long = 9
Private Const COL_番号 As Long = 11
Private Const COL_職員アルファベット As Long = 12
Private Const COL_職員名 As Long = 13
Private Const COL_スケジュール開始 As Long = 14

' 行位置の定数
Private Const ROW_ヘッダ As Long = 2
Private Const ROW_データ開始 As Long = 3

' ========================================
' メインプロシージャ
' ========================================
Sub 職員配置タイムスケジュール表()
    On Error GoTo ErrorHandler
    
    Application.ScreenUpdating = False
    Application.Calculation = xlCalculationManual
    
    Dim ws As Worksheet
    Dim 時間区切り As Integer
    
    Set ws = ActiveSheet
    
    ' 時間区切りの選択
    時間区切り = 時間区切り選択()
    If 時間区切り = 0 Then
        Application.ScreenUpdating = True
        Application.Calculation = xlCalculationAutomatic
        Exit Sub
    End If
    
    ' 設定情報の取得
    Dim 設定情報 As Object
    Set 設定情報 = CreateObject("Scripting.Dictionary")
    
    ' シート選択による情報取得または新規入力
    If シート選択情報取得(ws, 設定情報) = False Then
        Application.ScreenUpdating = True
        Application.Calculation = xlCalculationAutomatic
        Exit Sub
    End If
    
    ' 入力内容の検証
    Dim エラーメッセージ As String
    エラーメッセージ = 設定内容検証(設定情報)
    If エラーメッセージ <> "" Then
        MsgBox エラーメッセージ, vbCritical, "入力エラー"
        Application.ScreenUpdating = True
        Application.Calculation = xlCalculationAutomatic
        Exit Sub
    End If
    
    ' シートのクリア
    ws.Cells.Clear
    
    ' スケジュール表の作成
    Application.StatusBar = "スケジュール表を作成中..."
    Call シフトパターン表作成(ws, 設定情報, 時間区切り)
    Call 職員リスト表作成(ws, 設定情報)
    Call タイムスケジュール表作成(ws, 設定情報, 時間区切り)
    Call 作成ボタン配置(ws, 設定情報)
    Call ヘルプ表示領域作成(ws, 設定情報)
    
    Application.StatusBar = False
    Application.ScreenUpdating = True
    Application.Calculation = xlCalculationAutomatic
    
    MsgBox "スケジュール表のテンプレートが作成されました。" & vbNewLine & _
           "勤務時間とアルファベットを入力後、「作成」ボタンを押してください。", _
           vbInformation, "完了"
    
    Exit Sub
    
ErrorHandler:
    Application.ScreenUpdating = True
    Application.Calculation = xlCalculationAutomatic
    Application.StatusBar = False
    MsgBox "エラーが発生しました。" & vbNewLine & _
           "エラー番号: " & Err.Number & vbNewLine & _
           "エラー内容: " & Err.Description, vbCritical, "エラー"
End Sub

' ========================================
' 時間区切り選択
' ========================================
Private Function 時間区切り選択() As Integer
    Dim 入力値 As String
    Dim 時間区切り As Integer
    
    入力値 = InputBox("1セルにおける時間の単位(分)を選択してください" & vbNewLine & _
                     "※5, 15, 30のいずれかを入力", "時間単位設定", "5")
    
    If 入力値 = "" Then
        時間区切り選択 = 0
        Exit Function
    End If
    
    On Error Resume Next
    時間区切り = CInt(入力値)
    On Error GoTo 0
    
    If 時間区切り <> 5 And 時間区切り <> 15 And 時間区切り <> 30 Then
        MsgBox "無効な入力です。5分間隔に設定します。", vbInformation, "設定"
        時間区切り = 5
    End If
    
    時間区切り選択 = 時間区切り
End Function

' ========================================
' シート選択による情報取得
' ========================================
Private Function シート選択情報取得(ws As Worksheet, 設定情報 As Object) As Boolean
    ' 他のシートが存在するか確認
    If ThisWorkbook.Worksheets.Count = 1 Then
        ' 他のシートがない場合は新規入力
        シート選択情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
    
    ' 他のシートの情報を使用するか確認
    Dim 応答 As VbMsgBoxResult
    応答 = MsgBox("他のシートの情報を使用しますか？" & vbNewLine & _
                  "「はい」を選択すると、コピー元のシートを選択できます。", _
                  vbYesNo + vbQuestion, "確認")
    
    If 応答 = vbNo Then
        シート選択情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
    
    ' シート選択
    Dim 選択シート As Worksheet
    Set 選択シート = シート選択ダイアログ(ws)
    
    If 選択シート Is Nothing Then
        ' キャンセルの場合は新規入力
        シート選択情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
    
    ' 選択したシートから情報を取得
    シート選択情報取得 = 選択シート情報取得(選択シート, 設定情報)
End Function

' ========================================
' シート選択ダイアログ
' ========================================
Private Function シート選択ダイアログ(現在シート As Worksheet) As Worksheet
    On Error GoTo ErrorHandler
    
    Dim シートリスト As String
    Dim シート名配列() As String
    Dim i As Long
    Dim カウント As Long
    
    ' 現在のシート以外のシートリストを作成
    カウント = 0
    For i = 1 To ThisWorkbook.Worksheets.Count
        If ThisWorkbook.Worksheets(i).Name <> 現在シート.Name Then
            カウント = カウント + 1
            ReDim Preserve シート名配列(1 To カウント)
            シート名配列(カウント) = ThisWorkbook.Worksheets(i).Name
            シートリスト = シートリスト & カウント & ". " & ThisWorkbook.Worksheets(i).Name & vbNewLine
        End If
    Next i
    
    If カウント = 0 Then
        MsgBox "コピー可能なシートがありません。", vbInformation, "情報"
        Set シート選択ダイアログ = Nothing
        Exit Function
    End If
    
    ' シート番号の入力
    Dim 入力値 As String
    Dim 選択番号 As Long
    
    Do
        入力値 = InputBox("コピー元のシートを番号で選択してください：" & vbNewLine & vbNewLine & _
                         シートリスト & vbNewLine & _
                         "※キャンセルすると新規入力になります", "シート選択")
        
        If 入力値 = "" Then
            Set シート選択ダイアログ = Nothing
            Exit Function
        End If
        
        On Error Resume Next
        選択番号 = CLng(入力値)
        On Error GoTo 0
        
        If 選択番号 >= 1 And 選択番号 <= カウント Then
            Set シート選択ダイアログ = ThisWorkbook.Worksheets(シート名配列(選択番号))
            Exit Function
        Else
            MsgBox "1から" & カウント & "の番号を入力してください。", vbExclamation, "入力エラー"
        End If
    Loop
    
    Exit Function
    
ErrorHandler:
    MsgBox "シート選択中にエラーが発生しました。", vbExclamation, "エラー"
    Set シート選択ダイアログ = Nothing
End Function


' ========================================
' 選択シート情報取得
' ========================================
Private Function 選択シート情報取得(選択シート As Worksheet, 設定情報 As Object) As Boolean

Dim シフトパターン数 As Long
Dim 職員数 As Long
Dim i As Long

On Error GoTo ErrorHandler

' シフトパターン数の確認
If MsgBox("シフトのパターン数は選択したシートと同じですか？", vbYesNo + vbQuestion, "確認") = vbYes Then
    ' 実際にデータがある行をカウント(空白行を除外)
    シフトパターン数 = 0
    For i = ROW_データ開始 To 選択シート.Rows.Count
        Dim セル値 As String
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(選択シート.Cells(i, COL_アルファベット).Value))
        On Error GoTo ErrorHandler
        
        If セル値 <> "" Then
            シフトパターン数 = シフトパターン数 + 1
        ElseIf シフトパターン数 > 0 Then
            ' データの後に空白行が来たら終了
            Exit For
        End If
    Next i
    
    If シフトパターン数 = 0 Then
        MsgBox "選択したシートにシフトパターンが見つかりませんでした。" & vbNewLine & _
               "新規入力画面に進みます。", vbInformation, "通知"
        選択シート情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
Else
    シフトパターン数 = 数値入力("シフトのパターン数を入力してください", "シフトパターン数", 1)
    If シフトパターン数 = 0 Then
        選択シート情報取得 = False
        Exit Function
    End If
End If

設定情報("シフトパターン数") = シフトパターン数

' シフトパターン配列の初期化
Dim シフトパターン配列() As String
ReDim シフトパターン配列(1 To シフトパターン数)

' シフトのアルファベットと勤務時間の確認
If MsgBox("シフトのアルファベットは選択したシートと同じですか？", vbYesNo + vbQuestion, "確認") = vbYes Then
    ' 実際にデータがある行から取得
    Dim 取得カウント As Long
    取得カウント = 0
    
    For i = ROW_データ開始 To 選択シート.Rows.Count
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(選択シート.Cells(i, COL_アルファベット).Value))
        On Error GoTo ErrorHandler
        
        If セル値 <> "" Then
            取得カウント = 取得カウント + 1
            シフトパターン配列(取得カウント) = セル値
            If 取得カウント >= シフトパターン数 Then
                Exit For
            End If
        End If
    Next i
    
    ' 取得できたパターン数が不足している場合
    If 取得カウント < シフトパターン数 Then
        MsgBox "選択したシートのシフトパターンが不足しています。" & vbNewLine & _
               "新規入力画面に進みます。", vbInformation, "通知"
        選択シート情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
    
    設定情報("前回勤務時間使用") = True
    Set 設定情報("前シート") = 選択シート
Else
    For i = 1 To シフトパターン数
        シフトパターン配列(i) = InputBox(i & "番目のシフトのパターンのアルファベットを入力してください", _
                                          "シフトパターン " & i)
        If シフトパターン配列(i) = "" Then
            MsgBox "アルファベットが入力されませんでした。処理を中止します。", vbExclamation, "エラー"
            選択シート情報取得 = False
            Exit Function
        End If
    Next i
    
    設定情報("前回勤務時間使用") = False
End If

設定情報("シフトパターン配列") = シフトパターン配列

' 職員数の確認
If MsgBox("職員数は選択したシートと同じですか?", vbYesNo + vbQuestion, "確認") = vbYes Then
    職員数 = 0
    ' K列(COL_番号)のデータをカウント
    For i = ROW_データ開始 To 選択シート.Rows.Count
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(選択シート.Cells(i, COL_番号).Value))
        On Error GoTo ErrorHandler
        
        If セル値 <> "" Then
            職員数 = 職員数 + 1
        ElseIf 職員数 > 0 Then
            Exit For
        End If
    Next i
    
    If 職員数 = 0 Then
        MsgBox "選択したシートに職員データが見つかりませんでした。" & vbNewLine & _
               "新規入力画面に進みます。", vbInformation, "通知"
        選択シート情報取得 = 新規設定入力(設定情報)
        Exit Function
    End If
Else
    職員数 = 数値入力("職員数を入力してください", "職員数", 1)
    If 職員数 = 0 Then
        選択シート情報取得 = False
        Exit Function
    End If
End If

設定情報("職員数") = 職員数

' タイムスケジュール表の時間範囲の確認
Dim 開始時間 As Date
Dim 終了時間 As Date
Dim 時間取得成功 As Boolean

時間取得成功 = False

If MsgBox("タイムスケジュール表の時間は選択したシートと同じですか?", vbYesNo + vbQuestion, "確認") = vbYes Then
    ' 開始時間の取得を試みる
    開始時間 = 時刻セル値取得(選択シート, ROW_ヘッダ, COL_スケジュール開始)
    
    ' 終了時間の取得(最終列を探す)
    Dim 終了時間列 As Long
    On Error Resume Next
    終了時間列 = 選択シート.Cells(ROW_ヘッダ, 選択シート.Columns.Count).End(xlToLeft).Column
    On Error GoTo ErrorHandler
    
    If 終了時間列 > COL_スケジュール開始 Then
        終了時間 = 時刻セル値取得(選択シート, ROW_ヘッダ, 終了時間列)
    End If
    
    ' 時間が正しく取得できたかチェック
    If 開始時間 > 0 And 終了時間 > 0 And 終了時間 > 開始時間 Then
        時間取得成功 = True
    Else
        MsgBox "選択したシートから時間情報を正しく取得できませんでした。" & vbNewLine & _
               "手動で時間を入力してください。", vbInformation, "通知"
    End If
End If

' 時間取得に失敗した場合は手動入力
If Not 時間取得成功 Then
    If 時間入力(開始時間, 終了時間) = False Then
        選択シート情報取得 = False
        Exit Function
    End If
End If

設定情報("開始時間") = 開始時間
設定情報("終了時間") = 終了時間

選択シート情報取得 = True
Exit Function

ErrorHandler:
    MsgBox "選択したシートからの情報取得中にエラーが発生しました。" & vbNewLine & _
           "エラー内容: " & Err.Description & vbNewLine & _
           "新規入力画面に進みます。", vbInformation, "通知"
    選択シート情報取得 = 新規設定入力(設定情報)
End Function

' ========================================
' 時刻セル値取得
' ========================================
Private Function 時刻セル値取得(ws As Worksheet, 行 As Long, 列 As Long) As Date
    Dim セル値 As Variant
    Dim 結果 As Date
    
    結果 = 0
    
    On Error Resume Next
    セル値 = ws.Cells(行, 列).Value
    
    ' 空白チェック
    If IsEmpty(セル値) Or セル値 = "" Then
        時刻セル値取得 = 0
        Exit Function
    End If
    
    ' 日付型の場合
    If IsDate(セル値) Then
        結果 = CDate(セル値)
    ' 数値型の場合（シリアル値）
    ElseIf IsNumeric(セル値) Then
        Dim 数値 As Double
        数値 = CDbl(セル値)
        ' シリアル値が1未満の場合は時刻として扱う
        If 数値 >= 0 And 数値 < 1 Then
            結果 = CDate(数値)
        ElseIf 数値 >= 1 Then
            ' 1以上の場合は日付＋時刻なので、時刻部分のみ取得
            結果 = TimeSerial(Hour(数値), Minute(数値), 0)
        End If
    ' 文字列の場合
    Else
        Dim 文字列 As String
        文字列 = Trim(CStr(セル値))
        ' "hh:mm"形式の文字列を変換
        If InStr(文字列, ":") > 0 Then
            結果 = TimeValue(文字列)
        End If
    End If
    
    On Error GoTo 0
    
    時刻セル値取得 = 結果
End Function

' ========================================
' 新規設定入力
' ========================================
Private Function 新規設定入力(設定情報 As Object) As Boolean
    Dim シフトパターン数 As Long
    Dim 職員数 As Long
    Dim i As Long
    Dim 開始時間 As Date
    Dim 終了時間 As Date
    
    ' シフトパターン数の入力
    シフトパターン数 = 数値入力("シフトのパターン数を入力してください", "シフトパターン数", 1)
    If シフトパターン数 = 0 Then
        新規設定入力 = False
        Exit Function
    End If
    
    設定情報("シフトパターン数") = シフトパターン数
    
    ' シフトパターン配列の入力
    Dim シフトパターン配列() As String
    ReDim シフトパターン配列(1 To シフトパターン数)
    
    For i = 1 To シフトパターン数
        シフトパターン配列(i) = InputBox(i & "番目のシフトのパターンのアルファベットを入力してください", _
                                       "シフトパターン " & i)
        If シフトパターン配列(i) = "" Then
            MsgBox "アルファベットが入力されませんでした。処理を中止します。", vbExclamation, "エラー"
            新規設定入力 = False
            Exit Function
        End If
    Next i
    
    設定情報("シフトパターン配列") = シフトパターン配列
    設定情報("前回勤務時間使用") = False
    
    ' 職員数の入力
    職員数 = 数値入力("職員数を入力してください", "職員数", 1)
    If 職員数 = 0 Then
        新規設定入力 = False
        Exit Function
    End If
    
    設定情報("職員数") = 職員数
    
    ' 時間範囲の入力
    If 時間入力(開始時間, 終了時間) = False Then
        新規設定入力 = False
        Exit Function
    End If
    
    設定情報("開始時間") = 開始時間
    設定情報("終了時間") = 終了時間
    
    新規設定入力 = True
End Function

' ========================================
' 数値入力（検証付き）
' ========================================
Private Function 数値入力(プロンプト As String, タイトル As String, 最小値 As Long) As Long
    Dim 入力値 As String
    Dim 数値 As Long
    
    Do
        入力値 = InputBox(プロンプト, タイトル)
        
        If 入力値 = "" Then
            数値入力 = 0
            Exit Function
        End If
        
        On Error Resume Next
        数値 = CLng(入力値)
        On Error GoTo 0
        
        If 数値 < 最小値 Then
            MsgBox "1以上の数値を入力してください。", vbExclamation, "入力エラー"
        Else
            Exit Do
        End If
    Loop
    
    数値入力 = 数値
End Function

' ========================================
' 時間入力（検証付き）
' ========================================
Private Function 時間入力(開始時間 As Date, 終了時間 As Date) As Boolean
    Dim 開始時間文字列 As String
    Dim 終了時間文字列 As String
    
    Do
        開始時間文字列 = InputBox("タイムスケジュール表の開始時間を入力してください" & vbNewLine & _
                                "(例: 9:00 または 09:00)", "開始時間", "9:00")
        
        If 開始時間文字列 = "" Then
            時間入力 = False
            Exit Function
        End If
        
        On Error Resume Next
        開始時間 = TimeValue(開始時間文字列)
        If Err.Number <> 0 Then
            MsgBox "無効な時間形式です。「9:00」または「09:00」の形式で入力してください。", vbExclamation, "入力エラー"
            Err.Clear
        Else
            Exit Do
        End If
        On Error GoTo 0
    Loop
    
    Do
        終了時間文字列 = InputBox("タイムスケジュール表の終了時間を入力してください" & vbNewLine & _
                                "(例: 18:00)", "終了時間", "18:00")
        
        If 終了時間文字列 = "" Then
            時間入力 = False
            Exit Function
        End If
        
        On Error Resume Next
        終了時間 = TimeValue(終了時間文字列)
        If Err.Number <> 0 Then
            MsgBox "無効な時間形式です。「18:00」の形式で入力してください。", vbExclamation, "入力エラー"
            Err.Clear
        ElseIf 終了時間 <= 開始時間 Then
            MsgBox "終了時間は開始時間より後の時間を入力してください。" & vbNewLine & _
                   "開始時間: " & Format(開始時間, "hh:mm"), vbExclamation, "入力エラー"
            Err.Clear
        Else
            Exit Do
        End If
        On Error GoTo 0
    Loop
    
    時間入力 = True
End Function

' ========================================
' 設定内容検証
' ========================================
Private Function 設定内容検証(設定情報 As Object) As String
    Dim エラーメッセージ As String
    Dim シフトパターン配列() As String
    Dim i As Long
    
    シフトパターン配列 = 設定情報("シフトパターン配列")
    
    ' アルファベットの重複チェック
    Dim チェック用 As Object
    Set チェック用 = CreateObject("Scripting.Dictionary")
    
    For i = LBound(シフトパターン配列) To UBound(シフトパターン配列)
        If チェック用.Exists(シフトパターン配列(i)) Then
            エラーメッセージ = エラーメッセージ & "・アルファベット「" & シフトパターン配列(i) & _
                            "」が重複しています。" & vbNewLine
        Else
            チェック用.Add シフトパターン配列(i), True
        End If
        
        ' 空白チェック
        If Trim(シフトパターン配列(i)) = "" Then
            エラーメッセージ = エラーメッセージ & "・" & i & "番目のアルファベットが空白です。" & vbNewLine
        End If
    Next i
    
    If エラーメッセージ <> "" Then
        エラーメッセージ = "以下のエラーがあります：" & vbNewLine & vbNewLine & エラーメッセージ
    End If
    
    設定内容検証 = エラーメッセージ
End Function

' ========================================
' シフトパターン表作成
' ========================================
Private Sub シフトパターン表作成(ws As Worksheet, 設定情報 As Object, 時間区切り As Integer)
    Dim シフトパターン数 As Long
    Dim シフトパターン配列() As String
    Dim i As Long
    
    シフトパターン数 = 設定情報("シフトパターン数")
    シフトパターン配列 = 設定情報("シフトパターン配列")
    
    ' ヘッダの設定
    ws.Cells(ROW_ヘッダ, COL_アルファベット).Value = "アルファベット"
    ws.Cells(ROW_ヘッダ, COL_勤務開始時).Value = "勤務時間"
    ws.Range(ws.Cells(ROW_ヘッダ, COL_勤務開始時), ws.Cells(ROW_ヘッダ, COL_勤務終了分)).Merge
    ws.Range("B2:I2").HorizontalAlignment = xlCenter
    ws.Range("B2:I2").VerticalAlignment = xlCenter
    
    ' 列幅の設定
    ws.Columns("B:B").ColumnWidth = 15
    ws.Columns("C:C").ColumnWidth = 5
    ws.Columns("D:D").ColumnWidth = 2
    ws.Columns("E:E").ColumnWidth = 5
    ws.Columns("F:F").ColumnWidth = 2
    ws.Columns("G:G").ColumnWidth = 5
    ws.Columns("H:H").ColumnWidth = 2
    ws.Columns("I:I").ColumnWidth = 5
    
    ' 各シフトパターンの設定
    For i = 1 To シフトパターン数
        ws.Cells(ROW_データ開始 - 1 + i, COL_アルファベット).Value = シフトパターン配列(i)
        ' 前回の勤務時間を使用する場合
        If 設定情報("前回勤務時間使用") = True Then
            Dim prevWs As Worksheet
            Set prevWs = 設定情報("前シート")
            
            ' 元シートから該当するアルファベットの行を探す
            Dim 元シート行 As Long
            元シート行 = 0
            Dim j As Long
            For j = ROW_データ開始 To prevWs.Rows.Count
                Dim 元セル値 As String
                元セル値 = ""
                On Error Resume Next
                元セル値 = Trim(CStr(prevWs.Cells(j, COL_アルファベット).Value))
                On Error GoTo 0
                
                If 元セル値 = シフトパターン配列(i) Then
                    元シート行 = j
                    Exit For
                ElseIf 元セル値 = "" And j > ROW_データ開始 + シフトパターン数 Then
                    ' データ範囲を超えたら終了
                    Exit For
                End If
            Next j
            
            ' 見つかった場合のみコピー
            If 元シート行 > 0 Then
                On Error Resume Next
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務開始時).Value = prevWs.Cells(元シート行, COL_勤務開始時).Value
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務開始分).Value = prevWs.Cells(元シート行, COL_勤務開始分).Value
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了時).Value = prevWs.Cells(元シート行, COL_勤務終了時).Value
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了分).Value = prevWs.Cells(元シート行, COL_勤務終了分).Value
                On Error GoTo 0
            End If
        End If
        
        ' 勤務時間の入力リストの設定（時）
        With ws.Cells(ROW_データ開始 - 1 + i, COL_勤務開始時).Validation
            .Delete
            .Add Type:=xlValidateList, AlertStyle:=xlValidAlertStop, Operator:=xlBetween, _
                 Formula1:="0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23"
            .InCellDropdown = True
        End With
        
        With ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了時).Validation
            .Delete
            .Add Type:=xlValidateList, AlertStyle:=xlValidAlertStop, Operator:=xlBetween, _
                 Formula1:="0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18,19,20,21,22,23"
            .InCellDropdown = True
        End With
        
        ' 勤務時間の入力リストの設定（分）
        With ws.Cells(ROW_データ開始 - 1 + i, COL_勤務開始分).Validation
            .Delete
            .Add Type:=xlValidateList, AlertStyle:=xlValidAlertStop, Operator:=xlBetween, _
                 Formula1:="00,05,10,15,20,25,30,35,40,45,50,55"
            .InCellDropdown = True
        End With
        
        With ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了分).Validation
            .Delete
            .Add Type:=xlValidateList, AlertStyle:=xlValidAlertStop, Operator:=xlBetween, _
                 Formula1:="00,05,10,15,20,25,30,35,40,45,50,55"
            .InCellDropdown = True
        End With
        
        ' 区切り文字の設定
        ws.Cells(ROW_データ開始 - 1 + i, 4).Value = ":"
        ws.Cells(ROW_データ開始 - 1 + i, 6).Value = "～"
        ws.Cells(ROW_データ開始 - 1 + i, 8).Value = ":"
        
        ' 中央揃えの設定
        ws.Range(ws.Cells(ROW_データ開始 - 1 + i, COL_アルファベット), _
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了分)).HorizontalAlignment = xlCenter
        ws.Range(ws.Cells(ROW_データ開始 - 1 + i, COL_アルファベット), _
                ws.Cells(ROW_データ開始 - 1 + i, COL_勤務終了分)).VerticalAlignment = xlCenter
    Next i
    
    ' ボーダーの設定
    With ws.Range(ws.Cells(ROW_ヘッダ, COL_アルファベット), ws.Cells(ROW_ヘッダ + シフトパターン数, COL_勤務終了分))
        .Borders.LineStyle = xlContinuous
        .Borders.Weight = xlThin
        .BorderAround LineStyle:=xlContinuous, Weight:=xlMedium
    End With
End Sub

' ========================================
' 職員リスト表作成
' ========================================
Private Sub 職員リスト表作成(ws As Worksheet, 設定情報 As Object)
    Dim 職員数 As Long
    Dim i As Long
    
    職員数 = 設定情報("職員数")
    
    ' タイトルの設定
    ws.Cells(ROW_ヘッダ, COL_番号).Value = "番号"
    ws.Cells(ROW_ヘッダ, COL_職員アルファベット).Value = "アルファベット"
    ws.Cells(ROW_ヘッダ, COL_職員名).Value = "職員名"
    ws.Range("K2:M2").HorizontalAlignment = xlCenter
    ws.Range("K2:M2").VerticalAlignment = xlCenter
    
    ' 列幅の設定
    ws.Columns("K:K").ColumnWidth = 5
    ws.Columns("L:L").ColumnWidth = 15
    ws.Columns("M:M").ColumnWidth = 15
    
    ' 番号の割り振り（配列を使用して高速化）
    Dim 番号配列() As Variant
    ReDim 番号配列(1 To 職員数, 1 To 1)
    
    For i = 1 To 職員数
        番号配列(i, 1) = i
    Next i
    
    ws.Range(ws.Cells(ROW_データ開始, COL_番号), ws.Cells(ROW_データ開始 - 1 + 職員数, COL_番号)).Value = 番号配列
    ws.Range(ws.Cells(ROW_データ開始, COL_番号), ws.Cells(ROW_データ開始 - 1 + 職員数, COL_番号)).HorizontalAlignment = xlCenter
    ws.Range(ws.Cells(ROW_データ開始, COL_番号), ws.Cells(ROW_データ開始 - 1 + 職員数, COL_番号)).VerticalAlignment = xlCenter
End Sub

' ========================================
' タイムスケジュール表作成
' ========================================
Private Sub タイムスケジュール表作成(ws As Worksheet, 設定情報 As Object, 時間区切り As Integer)
    Dim 職員数 As Long
    Dim 開始時間 As Date
    Dim 終了時間 As Date
    Dim 最終行 As Long
    Dim 最終列 As Long
    
    職員数 = 設定情報("職員数")
    開始時間 = 設定情報("開始時間")
    終了時間 = 設定情報("終了時間")
    最終行 = ROW_ヘッダ + 職員数
    
    ' N列以降の幅を均等に設定（修正）
    ' まず列幅をリセット
    ws.Columns("N:XFD").ColumnWidth = 2.5
    
    ' 時間表示の設定
    Dim currentTime As Date
    currentTime = 開始時間
    Dim lastColumn As Long
    lastColumn = COL_スケジュール開始
    
    Do While currentTime < 終了時間 Or currentTime = 終了時間
        ' 時間表示（常に2セルを結合）
        ws.Cells(ROW_ヘッダ, lastColumn).Value = currentTime
        ws.Cells(ROW_ヘッダ, lastColumn).NumberFormat = "hh:mm"
        ws.Cells(ROW_ヘッダ, lastColumn).HorizontalAlignment = xlCenter
        ws.Cells(ROW_ヘッダ, lastColumn).VerticalAlignment = xlCenter
        ws.Range(ws.Cells(ROW_ヘッダ, lastColumn), ws.Cells(ROW_ヘッダ, lastColumn + 1)).Merge
        
        ' 結合したセルの幅を均等に設定（修正）
        ws.Columns(lastColumn).ColumnWidth = 2.5
        ws.Columns(lastColumn + 1).ColumnWidth = 2.5
        
        ' 次の時間へ
        currentTime = DateAdd("n", 時間区切り * 2, currentTime)
        lastColumn = lastColumn + 2
    Loop
    
    ' 最終時刻が表示されていない場合は追加
    If DateDiff("n", DateAdd("n", -時間区切り * 2, currentTime), 終了時間) > 0 Then
        ws.Cells(ROW_ヘッダ, lastColumn).Value = 終了時間
        ws.Cells(ROW_ヘッダ, lastColumn).NumberFormat = "hh:mm"
        ws.Cells(ROW_ヘッダ, lastColumn).HorizontalAlignment = xlCenter
        ws.Cells(ROW_ヘッダ, lastColumn).VerticalAlignment = xlCenter
        ws.Range(ws.Cells(ROW_ヘッダ, lastColumn), ws.Cells(ROW_ヘッダ, lastColumn + 1)).Merge
        
        ' 結合したセルの幅を均等に設定
        ws.Columns(lastColumn).ColumnWidth = 2.5
        ws.Columns(lastColumn + 1).ColumnWidth = 2.5
        
        lastColumn = lastColumn + 2
    End If
    
    最終列 = lastColumn - 1
    
    ' 使用した列すべての幅を最終的に均等に設定
    Dim col As Long
    For col = COL_スケジュール開始 To 最終列
        ws.Columns(col).ColumnWidth = 2.5
    Next col
    
    ' 罫線の設定
    With ws.Range(ws.Cells(ROW_ヘッダ, COL_番号), ws.Cells(最終行, 最終列))
        .Borders(xlEdgeTop).LineStyle = xlContinuous
        .Borders(xlEdgeTop).Weight = xlMedium
        .Borders(xlEdgeBottom).LineStyle = xlContinuous
        .Borders(xlEdgeBottom).Weight = xlMedium
        .Borders(xlEdgeLeft).LineStyle = xlContinuous
        .Borders(xlEdgeLeft).Weight = xlMedium
        .Borders(xlEdgeRight).LineStyle = xlContinuous
        .Borders(xlEdgeRight).Weight = xlMedium
        .Borders(xlInsideVertical).LineStyle = xlContinuous
        .Borders(xlInsideVertical).Weight = xlThin
        .Borders(xlInsideHorizontal).LineStyle = xlContinuous
        .Borders(xlInsideHorizontal).Weight = xlThin
    End With
    
    ' 外枠を太線に設定
    With ws.Range(ws.Cells(ROW_ヘッダ, COL_番号), ws.Cells(最終行, 最終列))
        .BorderAround LineStyle:=xlContinuous, Weight:=xlMedium
    End With
End Sub

' ========================================
' 作成ボタン配置
' ========================================
Private Sub 作成ボタン配置(ws As Worksheet, 設定情報 As Object)
    Dim 職員数 As Long
    Dim 最終行 As Long
    Dim ボタン位置 As Range
    Dim 作成ボタン As Shape
    
    職員数 = 設定情報("職員数")
    最終行 = ROW_ヘッダ + 職員数
    
    ' 既存のボタンを削除
    On Error Resume Next
    ws.Shapes("作成ボタン").Delete
    ws.Shapes("クリアボタン").Delete
    On Error GoTo 0
    
    ' 作成ボタン
    Set ボタン位置 = ws.Cells(最終行 + 2, COL_アルファベット)
    Set 作成ボタン = ws.Shapes.AddFormControl(xlButtonControl, ボタン位置.Left, ボタン位置.Top, 80, 30)
    
    With 作成ボタン
        .Name = "作成ボタン"
        .OnAction = "スケジュール作成"
        .TextFrame.Characters.Text = "作成"
    End With
    
    ' クリアボタン
    Dim クリアボタン位置 As Range
    Dim クリアボタン As Shape
    
    Set クリアボタン位置 = ws.Cells(最終行 + 2, COL_アルファベット + 2)
    Set クリアボタン = ws.Shapes.AddFormControl(xlButtonControl, クリアボタン位置.Left, クリアボタン位置.Top, 80, 30)
    
    With クリアボタン
        .Name = "クリアボタン"
        .OnAction = "スケジュールクリア"
        .TextFrame.Characters.Text = "クリア"
    End With
End Sub

' ========================================
' ヘルプ表示領域作成
' ========================================
Private Sub ヘルプ表示領域作成(ws As Worksheet, 設定情報 As Object)
    Dim 職員数 As Long
    Dim 最終行 As Long
    Dim ヘルプ開始行 As Long
    Dim ヘルプ範囲 As Range
    
    職員数 = 設定情報("職員数")
    最終行 = ROW_ヘッダ + 職員数
    ヘルプ開始行 = 最終行 + 4
    
    ' ヘルプ表示用のセル範囲を結合
    Set ヘルプ範囲 = ws.Range(ws.Cells(ヘルプ開始行, COL_アルファベット), ws.Cells(ヘルプ開始行 + 15, COL_勤務終了分))
    
    ' セルを結合
    On Error Resume Next
    ヘルプ範囲.UnMerge
    On Error GoTo 0
    ヘルプ範囲.Merge
    
    ' ヘルプテキストの設定
    Dim ヘルプテキスト As String
    ヘルプテキスト = "【使い方ガイド】" & vbNewLine & vbNewLine & _
                    "■基本的な使い方" & vbNewLine & _
                    "1. 左側の「アルファベット」欄でシフトの種類を確認" & vbNewLine & _
                    "2. 各シフトの「勤務時間」を設定（ドロップダウンから選択）" & vbNewLine & _
                    "3. 右側の表で各職員に「アルファベット」を割り当て" & vbNewLine & _
                    "4. 「作成」ボタンをクリックしてスケジュールを表示" & vbNewLine & vbNewLine & _
                    "■便利な機能" & vbNewLine & _
                    "・「クリア」ボタン: スケジュールの色付けのみをクリア" & vbNewLine & _
                    "  （設定内容は残るので、修正後に再作成可能）" & vbNewLine & vbNewLine & _
                    "■時間設定のルール" & vbNewLine & _
                    "・開始時間と終了時間は表の範囲内に設定" & vbNewLine & _
                    "・終了時間は開始時間より後の時間を指定" & vbNewLine & _
                    "・時間と分の両方を入力してください" & vbNewLine & vbNewLine & _
                    "■エラーが出た場合" & vbNewLine & _
                    "・赤く表示されたセルを確認して修正" & vbNewLine & _
                    "・エラーメッセージの指示に従って入力" & vbNewLine & _
                    "・修正後、再度「作成」ボタンをクリック"
    
    ' セルにテキストを設定
    With ヘルプ範囲
        .Value = ヘルプテキスト
        .HorizontalAlignment = xlLeft
        .VerticalAlignment = xlTop
        .WrapText = True
        .Font.Size = 10
        .Font.Name = "メイリオ"
        .Interior.Color = RGB(255, 250, 205) ' 薄い黄色
        .Borders.LineStyle = xlContinuous
        .Borders.Weight = xlMedium
        .Borders.Color = RGB(255, 165, 0) ' オレンジ色の枠線
    End With
    
    ' 行の高さを自動調整
    ws.Rows(ヘルプ開始行 & ":" & ヘルプ開始行 + 15).RowHeight = 18
End Sub

' ========================================
' スケジュール作成
' ========================================
Sub スケジュール作成()
    On Error GoTo ErrorHandler
    
    Application.ScreenUpdating = False
    
    Dim ws As Worksheet
    Set ws = ActiveSheet
    
    Dim i As Long, j As Long
    Dim シフトパターン数 As Long, 職員数 As Long
    Dim アルファベット As String
    Dim 開始時間 As Date, 終了時間 As Date
    Dim エラーメッセージ As String
    Dim 有効なアルファベット As Object
    Dim 最終行 As Long
    Dim 最終列 As Long
    Dim 時間区切り As Integer
    
    ' プログレスバー表示
    Application.StatusBar = "スケジュール作成を開始しています..."
    
    ' シフトパターン数と職員数を取得（実際にデータがある行のみカウント）
    シフトパターン数 = 0
    For i = ROW_データ開始 To ws.Rows.Count
        Dim セル値 As String
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(ws.Cells(i, COL_アルファベット).Value))
        On Error GoTo 0
        
        If セル値 <> "" Then
            シフトパターン数 = シフトパターン数 + 1
        ElseIf シフトパターン数 > 0 Then
            Exit For
        End If
    Next i
    
    職員数 = 0
    For i = ROW_データ開始 To ws.Rows.Count
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(ws.Cells(i, COL_番号).Value))
        On Error GoTo 0
        
        If セル値 <> "" Then
            職員数 = 職員数 + 1
        ElseIf 職員数 > 0 Then
            Exit For
        End If
    Next i
    
    If シフトパターン数 = 0 Then
        MsgBox "シフトパターンが見つかりません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    If 職員数 = 0 Then
        MsgBox "職員データが見つかりません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    ' 開始時間を取得（2つ目のコードの方式）
    On Error Resume Next
    開始時間 = CDate(ws.Cells(ROW_ヘッダ, COL_スケジュール開始).Value)
    On Error GoTo 0
    
    If 開始時間 = 0 Then
        MsgBox "開始時間情報が正しく取得できません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    ' 次の時間ヘッダーを取得して時間区切りを計算
    Dim 次の時間 As Date
    On Error Resume Next
    次の時間 = CDate(ws.Cells(ROW_ヘッダ, COL_スケジュール開始 + 2).Value)
    On Error GoTo 0
    
    If 次の時間 = 0 Then
        MsgBox "時間情報が正しく取得できません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    ' 時間区切りを計算
    時間区切り = DateDiff("n", 開始時間, 次の時間) / 2
    
    If 時間区切り <= 0 Then
        MsgBox "時間区切りが正しく計算できません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    Set 有効なアルファベット = CreateObject("Scripting.Dictionary")
    
    ' 有効なアルファベットのみを登録（空白行をスキップ）
    For i = ROW_データ開始 To ROW_データ開始 + シフトパターン数 - 1
        Dim パターンアルファベット As String
        パターンアルファベット = ""
        On Error Resume Next
        パターンアルファベット = Trim(CStr(ws.Cells(i, COL_アルファベット).Value))
        On Error GoTo 0
        
        If パターンアルファベット <> "" Then
            有効なアルファベット(パターンアルファベット) = True
        End If
    Next i
    
    ' 表の終了時間を取得（最終列）
    Dim 終了時間列 As Long
    On Error Resume Next
    終了時間列 = ws.Cells(ROW_ヘッダ, ws.Columns.Count).End(xlToLeft).Column
    終了時間 = CDate(ws.Cells(ROW_ヘッダ, 終了時間列).Value)
    On Error GoTo 0
    
    If 終了時間 = 0 Then
        MsgBox "終了時間が正しく取得できません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        Application.StatusBar = False
        Exit Sub
    End If
    
    ' エラーチェック：勤務時間が入力されているか（アルファベットがある行のみチェック）
    For i = ROW_データ開始 To ROW_データ開始 + シフトパターン数 - 1
        ' アルファベットが空の行はスキップ
        セル値 = ""
        On Error Resume Next
        セル値 = Trim(CStr(ws.Cells(i, COL_アルファベット).Value))
        On Error GoTo 0
        
        If セル値 = "" Then
            GoTo NextPattern
        End If
        
        Dim 勤務開始時 As Variant, 勤務開始分 As Variant
        Dim 勤務終了時 As Variant, 勤務終了分 As Variant
        
        勤務開始時 = ws.Cells(i, COL_勤務開始時).Value
        勤務開始分 = ws.Cells(i, COL_勤務開始分).Value
        勤務終了時 = ws.Cells(i, COL_勤務終了時).Value
        勤務終了分 = ws.Cells(i, COL_勤務終了分).Value
        
        If Trim(CStr(勤務開始時)) = "" Or Trim(CStr(勤務開始分)) = "" Or _
           Trim(CStr(勤務終了時)) = "" Or Trim(CStr(勤務終了分)) = "" Then
            エラーメッセージ = エラーメッセージ & "【エラー】シフト「" & _
                            ws.Cells(i, COL_アルファベット).Value & _
                            "」の勤務時間が未入力です。" & vbNewLine & _
                            "→ 開始時間と終了時間を入力してください。" & vbNewLine & vbNewLine
        Else
            Dim 勤務開始時間 As Date, 勤務終了時間 As Date
            Dim 時間変換成功 As Boolean
            
            時間変換成功 = True
            On Error Resume Next
            勤務開始時間 = TimeSerial(CLng(勤務開始時), CLng(勤務開始分), 0)
            If Err.Number <> 0 Then
                時間変換成功 = False
            End If
            勤務終了時間 = TimeSerial(CLng(勤務終了時), CLng(勤務終了分), 0)
            If Err.Number <> 0 Then
                時間変換成功 = False
            End If
            On Error GoTo 0
            
            If Not 時間変換成功 Then
                エラーメッセージ = エラーメッセージ & "【エラー】シフト「" & _
                                ws.Cells(i, COL_アルファベット).Value & _
                                "」の勤務時間の形式が不正です。" & vbNewLine & vbNewLine
            Else
                ' 範囲外チェック
                If 勤務開始時間 < 開始時間 Or 勤務終了時間 > 終了時間 Then
                    エラーメッセージ = エラーメッセージ & "【エラー】シフト「" & _
                                    ws.Cells(i, COL_アルファベット).Value & _
                                    "」の勤務時間が表の範囲外です。" & vbNewLine & _
                                    "・設定: " & Format(勤務開始時間, "hh:mm") & " ～ " & _
                                    Format(勤務終了時間, "hh:mm") & vbNewLine & _
                                    "・表の範囲: " & Format(開始時間, "hh:mm") & " ～ " & _
                                    Format(終了時間, "hh:mm") & vbNewLine & _
                                    "→ 勤務時間を表の範囲内に修正してください。" & vbNewLine & vbNewLine
                    
                    ' エラーセルを強調表示
                    ws.Range(ws.Cells(i, COL_勤務開始時), _
                            ws.Cells(i, COL_勤務終了分)).Interior.Color = RGB(255, 200, 200)
                End If
                
                ' 開始時間≧終了時間のチェック
                If 勤務開始時間 >= 勤務終了時間 Then
                    エラーメッセージ = エラーメッセージ & "【エラー】シフト「" & _
                                    ws.Cells(i, COL_アルファベット).Value & _
                                    "」の終了時間が開始時間より前です。" & vbNewLine & _
                                    "・設定: " & Format(勤務開始時間, "hh:mm") & " ～ " & _
                                    Format(勤務終了時間, "hh:mm") & vbNewLine & _
                                    "→ 終了時間を開始時間より後に設定してください。" & vbNewLine & vbNewLine
                    
                    ' エラーセルを強調表示
                    ws.Range(ws.Cells(i, COL_勤務開始時), _
                            ws.Cells(i, COL_勤務終了分)).Interior.Color = RGB(255, 200, 200)
                End If
            End If
        End If
        
NextPattern:
    Next i
    
    ' 最終行と最終列を取得
    最終行 = ws.Cells(ws.Rows.Count, COL_番号).End(xlUp).Row
    最終列 = ws.Cells(ROW_ヘッダ, ws.Columns.Count).End(xlToLeft).Column
    
    ' パステルカラーの配列を定義
    Dim パステルカラー(7) As Long
    パステルカラー(0) = RGB(255, 182, 193) ' ライトピンク
    パステルカラー(1) = RGB(173, 216, 230) ' ライトブルー
    パステルカラー(2) = RGB(144, 238, 144) ' ライトグリーン
    パステルカラー(3) = RGB(255, 255, 224) ' ライトイエロー
    パステルカラー(4) = RGB(230, 230, 250) ' ラベンダー
    パステルカラー(5) = RGB(255, 218, 185) ' ピーチ
    パステルカラー(6) = RGB(176, 224, 230) ' パウダーブルー
    パステルカラー(7) = RGB(255, 192, 203) ' ピンク
    
    ' アルファベットと色のマッピングを作成
    Dim アルファベット色マップ As Object
    Set アルファベット色マップ = CreateObject("Scripting.Dictionary")
    Dim 色インデックス As Integer
    色インデックス = 0
    
    ' 既存のスケジュール範囲をクリア
    Dim スケジュール範囲 As Range
    Set スケジュール範囲 = ws.Range(ws.Cells(ROW_データ開始, COL_スケジュール開始), ws.Cells(最終行, 最終列))
    スケジュール範囲.Interior.ColorIndex = xlNone
    
    ' スケジュールの作成
    For i = 1 To 職員数
        Application.StatusBar = "スケジュール作成中... " & Format(i / 職員数, "0%")
        
        アルファベット = ""
        On Error Resume Next
        アルファベット = Trim(CStr(ws.Cells(ROW_データ開始 - 1 + i, COL_職員アルファベット).Value))
        On Error GoTo 0
        
        If アルファベット <> "" Then
            If Not 有効なアルファベット.Exists(アルファベット) Then
                エラーメッセージ = エラーメッセージ & "【エラー】職員番号 " & i & " のアルファベット「" & アルファベット & _
                                "」は左側の勤務表に存在しません。" & vbNewLine & _
                                "→ 正しいアルファベットを入力してください。" & vbNewLine
                
                If 有効なアルファベット.Count > 0 Then
                    エラーメッセージ = エラーメッセージ & "（利用可能: " & Join(有効なアルファベット.Keys, ", ") & "）" & vbNewLine & vbNewLine
                Else
                    エラーメッセージ = エラーメッセージ & vbNewLine
                End If
                
                ' エラーセルを強調表示
                ws.Cells(ROW_データ開始 - 1 + i, COL_職員アルファベット).Interior.Color = RGB(255, 200, 200)
            Else
                Dim シフト行 As Long
                シフト行 = 0
                
                ' アルファベットが一致する行を探す
                For j = ROW_データ開始 To ROW_データ開始 + シフトパターン数 - 1
                    Dim 比較セル値 As String
                    比較セル値 = ""
                    On Error Resume Next
                    比較セル値 = Trim(CStr(ws.Cells(j, COL_アルファベット).Value))
                    On Error GoTo 0
                    
                    If 比較セル値 = アルファベット Then
                        シフト行 = j
                        Exit For
                    End If
                Next j
                
                If シフト行 > 0 Then
                    勤務開始時 = ws.Cells(シフト行, COL_勤務開始時).Value
                    勤務開始分 = ws.Cells(シフト行, COL_勤務開始分).Value
                    勤務終了時 = ws.Cells(シフト行, COL_勤務終了時).Value
                    勤務終了分 = ws.Cells(シフト行, COL_勤務終了分).Value
                    
                    ' 勤務時間が全て入力されている場合のみ処理
                    If Trim(CStr(勤務開始時)) <> "" And Trim(CStr(勤務開始分)) <> "" And _
                       Trim(CStr(勤務終了時)) <> "" And Trim(CStr(勤務終了分)) <> "" Then
                        
                        On Error Resume Next
                        勤務開始時間 = TimeSerial(CLng(勤務開始時), CLng(勤務開始分), 0)
                        勤務終了時間 = TimeSerial(CLng(勤務終了時), CLng(勤務終了分), 0)
                        
                        If Err.Number = 0 Then
                            ' 開始時間からの経過分数を計算
                            Dim 開始セル As Long, 終了セル As Long
                            開始セル = (勤務開始時間 - 開始時間) * 24 * 60 / 時間区切り + COL_スケジュール開始 + 1
                            終了セル = (勤務終了時間 - 開始時間) * 24 * 60 / 時間区切り + COL_スケジュール開始
                            
                            ' 範囲チェック
                            If 開始セル >= COL_スケジュール開始 And 終了セル <= 最終列 And 開始セル <= 終了セル Then
                                ' アルファベットに対応する色を取得または割り当て
                                If Not アルファベット色マップ.Exists(アルファベット) Then
                                    アルファベット色マップ(アルファベット) = パステルカラー(色インデックス)
                                    色インデックス = (色インデックス + 1) Mod (UBound(パステルカラー) + 1)
                                End If
                                
                                ' 帯の表示（アルファベットに対応するパステルカラー）
                                With ws.Range(ws.Cells(ROW_データ開始 - 1 + i, 開始セル), ws.Cells(ROW_データ開始 - 1 + i, 終了セル))
                                    .Interior.Color = アルファベット色マップ(アルファベット)
                                    .Borders.LineStyle = xlContinuous
                                    .Borders.Weight = xlThin
                                End With
                            End If
                        End If
                        Err.Clear
                        On Error GoTo 0
                    End If
                End If
            End If
        End If
    Next i
    
    ' 外枠を太線に設定（スケジュール作成後）
    With ws.Range(ws.Cells(ROW_ヘッダ, COL_番号), ws.Cells(最終行, 最終列))
        .BorderAround LineStyle:=xlContinuous, Weight:=xlMedium
    End With
    
    Application.StatusBar = False
    Application.ScreenUpdating = True
    
    If エラーメッセージ <> "" Then
        MsgBox "以下のエラーがあります。修正後、再度「作成」ボタンを押してください。" & vbNewLine & vbNewLine & _
               エラーメッセージ, vbExclamation, "エラー"
    Else
        MsgBox "スケジュールが作成されました。", vbInformation, "完了"
    End If
    
    Exit Sub
    
ErrorHandler:
    Application.ScreenUpdating = True
    Application.StatusBar = False
    MsgBox "スケジュール作成中にエラーが発生しました。" & vbNewLine & _
           "エラー番号: " & Err.Number & vbNewLine & _
           "エラー内容: " & Err.Description, vbCritical, "エラー"
End Sub

' ========================================
' スケジュールクリア
' ========================================
Sub スケジュールクリア()
    On Error GoTo ErrorHandler
    
    Dim ws As Worksheet
    Set ws = ActiveSheet
    
    If MsgBox("スケジュールの色付けをクリアしますか？" & vbNewLine & _
              "（設定内容は残ります）", vbYesNo + vbQuestion, "確認") = vbNo Then
        Exit Sub
    End If
    
    Application.ScreenUpdating = False
    
    Dim 最終行 As Long
    Dim 最終列 As Long
    
    ' 最終行と最終列の取得（エラー処理付き）
    On Error Resume Next
    最終行 = ws.Cells(ws.Rows.Count, COL_番号).End(xlUp).Row
    最終列 = ws.Cells(ROW_ヘッダ, ws.Columns.Count).End(xlToLeft).Column
    
    If Err.Number <> 0 Or 最終行 < ROW_ヘッダ Or 最終列 < COL_番号 Then
        MsgBox "スケジュール範囲を特定できません。", vbExclamation, "エラー"
        Application.ScreenUpdating = True
        On Error GoTo 0
        Exit Sub
    End If
    On Error GoTo 0
    
    ' スケジュール範囲の色をクリア
    Dim スケジュール範囲 As Range
    Set スケジュール範囲 = ws.Range(ws.Cells(ROW_データ開始, COL_スケジュール開始), ws.Cells(最終行, 最終列))
    スケジュール範囲.Interior.ColorIndex = xlNone
    
    ' エラー表示の赤色もクリア
    On Error Resume Next
    ws.Range(ws.Cells(ROW_データ開始, COL_アルファベット), ws.Cells(最終行, COL_職員アルファベット)).Interior.ColorIndex = xlNone
    On Error GoTo 0
    
    ' 罫線を再設定
    With スケジュール範囲
        .Borders.LineStyle = xlContinuous
        .Borders.Weight = xlThin
    End With
    
    With ws.Range(ws.Cells(ROW_ヘッダ, COL_番号), ws.Cells(最終行, 最終列))
        .BorderAround LineStyle:=xlContinuous, Weight:=xlMedium
    End With
    
    Application.ScreenUpdating = True
    
    MsgBox "スケジュールをクリアしました。", vbInformation, "完了"
    
    Exit Sub
    
ErrorHandler:
    Application.ScreenUpdating = True
    MsgBox "クリア中にエラーが発生しました。" & vbNewLine & _
           "エラー番号: " & Err.Number & vbNewLine & _
           "エラー内容: " & Err.Description, vbCritical, "エラー"
End Sub
